

PROJECTS—HI-FI—COMPUTERS

\$1.00 ■ JULY 1978

Radio-Electronics

THE MAGAZINE FOR NEW IDEAS IN ELECTRONICS

part 2: phone accessory
BUILD DIGI-TOLL
keep track of call costs


build full featured
DIGITAL DARKROOM TIMER
controls enlarger and safelight

which type is best
DISPOSABLE BATTERIES
carbon, alkaline, mercury

digital comes to hi-fi
DIGITAL HI-FI RECORDING
learn how it's done

days, weeks, months
LONG DELAY TIMER IC'S
for precision delays

PLUS: ★ Picture Tube Brighteners ★ Build Programmable Beeper
★ All About Audio Oscillators ★ CET Exam ★ Computer Corner
★ Hi-Fi Lab Test Reports ★ Hobby Corner ★ Service Clinic



A
GERNSBACK
PUBLICATION





Computer Program

Know what your long-distance calls cost as you make them

FRED BLECHMAN, K6UGT

WOULD YOU LIKE TO KNOW EXACTLY WHAT YOUR TOLL AND long-distance calls are costing you, *as you are making them*? With a simple computer program, you can synchronize yourself with the phone company billing machine, and have your computer screen print out your accumulated charge, and the time (in seconds) to the next added charge!

I used a Radio Shack TRS-80 computer, so the program is written in Radio-Shack Level 1 BASIC. If you have a different computer, you should be able to adapt it with no trouble. The program is written so that it allows the operator to request information on rates and discount periods, or bypass these instructions and go right into entering the three variables that determine the cost of the call: the initial time period, the initial time charge and the additional charge per minute. Once this data is entered, the operator places the phone call in the regular manner, and presses the ENTER key on the computer keyboard



ACTUAL SCREEN PHOTOS of program as it is running during a real phone call. Display tells you actual cost of call moment by moment.

when the party at the other end picks up their phone. That synchronizes your computer with the phone company timer—with absolutely **no connection to the phone lines!** From that point on, the computer display shows the initial cost and time period, counts down the seconds to the next added charge and (after the initial time period) displays the total time and cost continuously. To make the total cost more readable, it is enclosed in a graphic rectangle.

You must account for the operating speed of the computer. This is done in lines 130 and 140 by delaying the start of the countdown a few seconds, since it takes time for the computer to get to this point in the program on the first pass (mostly graphic delay). Also, I found my TRS-80 ran a FOR-NEXT loop at a rate of 490 loops-per-second, as shown in line 225. You might have to change this number slightly if your computer runs a bit faster or slower. This program uses 2495 bytes of memory of the 3583 available in the TRS-80 4K memory unit.

Telephone Toll Totalizer Program

© 1978 Fred Blechman

```

5 REM ★ PROGRAMMED FOR TRS-80 LEVEL I BASIC ★
10 REM ★ COPYRIGHT 1978 FRED BLECHMAN ★
15 REM ★ THREE LINES AFFECT THE TIMING OF THIS PROGRAM ★
16 REM ★ LINES 130 AND 140 DETERMINE THE STARTING COUNT ★
17 REM ★ (THIS LETS YOU CORRECT FOR THE GRAPHIC DELAY) ★
18 REM ★ LINE 225 CONTROLS THE SECONDS COUNTING ACCURACY ★
20 CLS:P.:P.P.
21 P." TELEPHONE TOLL TOTALIZER":P.:P.
25 IN."DO YOU WANT SPECIFIC INSTRUCTIONS? YES=1,NO=2":A
26 IF A=1 GOTO 300
28 CLS:P.:P.:P.
29 P." TELEPHONE TOLL TOTALIZER":P.:P.
30 IN."WHAT IS THE INITIAL TIME PERIOD (MINUTES)":P
35 P.:P.
40 IN."WHAT IS THE INITIAL CHARGE (CENTS)":I
45 P.:P.
50 IN."WHAT IS THE ADDITIONAL CHARGE PER MINUTE (CENTS)":M
51 CLS:P.:P.:P.:P.
55 P."WHEN THE PARTY AT THE OTHER END PICKS UP THE RECEIVER"
57 IN."PRESS ENTER TO START TIMING. . . ":A$
60 CLS:P.:P.
70 P."THE INITIAL COST OF THE FIRST":P;"MINUTE(S) IS:$":I/100
80 P." THE ADDITIONAL COST PER MINUTE IS:$":M/100
90 P.AT 847,"PRESS BREAK TO STOP COUNTING."
100 C=I/100:D=1
110 FOR X=22 TO 93
115 SET (X,22):SET(X,28):NEXT X
120 FOR Y=22 TO 28
125 SET(22,Y):SET(93,Y):NEXT Y
130 T=56
140 IF P=3 THEN T=176
150 GOTO 221
200 P.AT 526,"THE TOTAL CHARGE IS NOW:$":C
210 SET(93,24):SET(93,25):SET(93,26)
215 P.AT 651,"TOTAL TIME CHARGED IS NOW:";P+D;"MINUTES"
220 T=59:D=D+1
221 C=C+(M/100)
225 FOR X=1 TO 490: NEXT X
230 P.AT 330,"SECONDS TO THE NEXT ADDITIONAL CHARGE:";T
240 T=T-1
250 IF T=-1 GOTO 200
255 GOTO 225
300 CLS:P:"YOUR CHARGES ARE BASED UPON THREE THINGS:"
310 P." (1) INITIAL TIME PERIOD (1 or 3 MINUTES)"
320 P." (2) INITIAL CHARGE (FOR THE INITIAL PERIOD)"
330 P." (3) ADDITIONAL CHARGE PER MINUTE AFTER INITIAL PERIOD"
340 P.:P."IF YOU USE AN OPERATOR TO ASSIST YOU, THE INITIAL"
350 P.:P."TIME PERIOD IS 3 MINUTES. DIRECT DIAL IS 1 MINUTE."
360 P.:P."THE CHARGES ARE BASED ON THE DESTINATION CALLED. . . ."
370 P." . . . THESE ARE USUALLY LISTED IN THE FRONT OF THE PHONE BOOK"
380 P." . . . OR. . . . CALL OPERATOR FOR THE RATES."
390 P."DO YOU WANT INFORMATION ON DISCOUNT PERIODS?"
400 P.:IN:"YES=1,NO=2":B
410 IF B=2 GOTO 28
420 CLS:P:"THERE ARE TWO DISCOUNT RATES IN THE CONT. U.S.A.:":
430 P." 35% DISCOUNT: 5PM-11PM SUNDAY-FRIDAY"
440 P." 8AM-11PM HOLIDAYS":P.
450 P." 65% DISCOUNT: 11PM-8AM EVERY NIGHT"
460 P." 8AM-11PM SATURDAY"
470 P." 8AM-5PM SUNDAY"
475 P.:P."CHARGES ARE BASED ON TIME AT CALLING POINT!"
476 P.:P.:P.
480 IN."PRESS ENTER TO INPUT TIME AND CHARGE DATA. . . ":A$
490 GOTO 28

```